
Changelog
    

Note Styler's only public API is its usage as a formatter plugin. While you're welcome to play with its internals,
they can and will change without that change being reflected in Styler's semantic version.

  
    
  
  main



  
    
  
  1.8.0



  
    
  
  Improvements


Rewrite single-clause case statements to be assignments (h/t 🤖)
# before
case foo |> Bar.baz() |> Bop.boop() do
  {:ok, widget} ->
    x = y
    wodget(widget)
end

# after
{:ok, widget} = foo |> Bar.baz() |> Bop.boop()
x = y
wodget(widget)

  
    
  
  1.7.0


Surprising how fast numbers go up when you're following semver.
Two new features, one being a pipe optimization and the other a style-consistency-enforcer in cond statements.

  
    
  
  Improvements


	|> Enum.filter(fun) |> List.first([default]) => |> Enum.find([default], fun) (#242, h/t @janpieper)

cond
If the last clause's left-hand-side is a truthy atom, map literal, or tuple, rewrite it to be true
# before
cond do
  a -> b
  c -> d
  :else -> e
end

# styled
cond do
  a -> b
  c -> d
  true -> e
end
This also helps Styler identify 2-clause conds that can be rewritten to if/else more readily, like the following:
# before
cond do
  a -> b
  :else -> c
end

# styled
if a do
  b
else
  c
end

  
    
  
  1.6.0


That's right, a feature release again so soon!

  
    
  
  Improvements


This version of Styler adds many readability improvements around ExUnit assert and refute, specifically when working with 1. negations or 2. some Enum stdlib functions.
Some of these rewrites are not semantically equivalent due to refute passing for both nil and false.
ExUnit assert/refute rewrites
Styler now inverts negated (!, not) assert/refute (eg assert !x => refute x) statements, and further inverts refute with boolean comparison operators (refute x < y => assert x >= y) because non-trivial refutes are harder to reason about [ citation needed ]. Asserting something is not nil is the same as just asserting that something, so that's gone too now.
These changes are best summarized by the following table:
	before	styled
	assert !x	refute x
	assert not x	refute x
	assert !!x	assert x
	assert x != nil	assert x
	assert x == nil	no change
	assert is_nil(x)	no change
	assert !is_nil(x)	assert x
	assert x not in y	refute x in y
	refute negated	
	refute x	no change
	refute !x	assert x
	refute not x	assert x
	refute x != y	assert x == y
	refute x !== y	assert x === y
	refute x != nil	assert x == nil
	refute x not in y	assert x in y
	refute comparison	
	refute x < y	assert x >= y
	refute x <= y	assert x > y
	refute x > y	assert x <= y
	refute x >= y	assert x < y

	assert Enum.member?(y, x) -> assert x in y
	assert Enum.find(x, y) -> assert Enum.any?(x, y) (nb. not semantically equivalent in theory, but equivalent in practice)
	assert Enum.any?(y, & &1 == x) -> assert x in y
	assert Enum.any?(y, fn var -> var == x end) -> assert x in y


  
    
  
  Fixes


	alias lifting: fix bug lifting in snippets with a single ast node at the root level (like a credo config file) (#240, h/t @defndaines)


  
    
  
  1.5.1



  
    
  
  Fixes


	alias lifting: handle comments in snippets with no existing directives (#239, h/t @kerryb)


  
    
  
  1.5.0



  
    
  
  Improvements


	apply aliases to code. if a module is aliased, and then later referenced with its full name, Styler will now shorten it to its alias. (#235, h/t me)
	added :minimum_supported_elixir_version configuration to better support libraries using Styler (#231, h/t @maennchen)
	# styler:sort will now sort keys for struct/map typespecs (#213, h/t @rojnwa)


  
    
  
  Fixes


	apply alias lifting to snippets with no modules or module directives in them. (#189, @h/t @halfdan)
	fix de-sugaring of syntax-sugared keyword lists whose values weren't atoms in map values (#236, h/t @RisPNG)
	fix mix config sorting mangling floating comment blocks in some cases (#230 again, h/t @ryoung786)


  
    
  
  1.4.2



  
    
  
  Fixes


	fix comment misplacement for large comment blocks in config files and # styler:sort (#230, h/t @cschmatzler)


  
    
  
  1.4.1



  
    
  
  Improvements


	to_timeout/1 rewrites to use the next largest unit in some simple instances
  # before
  to_timeout(second: 60 * m)
  to_timeout(day: 7)
  # after
  to_timeout(minute: m)
  to_timeout(week: 1)



  
    
  
  Fixes


	fixed styler raising when encountering invalid function definition ast


  
    
  
  1.4.0


	A very nice change in alias lifting means Styler will make sure that your code is using the aliases that it's specified.
	Shoutout to the smartrent folks for finding pipifying recursion issues
	Elixir 1.17 improvements and fixes
	Elixir 1.19-dev: delete struct updates

Read on for details.

  
    
  
  Improvements


Alias Lifting
This release taught Styler to try just that little bit harder when doing alias lifting.
	general improvements around conflict detection, lifting in more correct places and fewer incorrect places (#193, h/t @jsw800)

	use knowledge of existing aliases to shorten invocations (#201, h/t me)
  example:
  alias A.B.C

  A.B.C.foo()
  A.B.C.bar()
  A.B.C.baz()
  becomes:
  alias A.B.C

  C.foo()
  C.bar()
  C.baz()


Struct Updates => Map Updates
1.19 deprecates struct update syntax in favor of map update syntax.
# This
%Struct{x | y}
# Styles to this
%{x | y}
WARNING Double check your diffs to make sure your variable is pattern matching against the same struct if you want to harness 1.19's type checking features. Apologies to folks who hoped Styler would do this step for you <3 (#199, h/t @SteffenDE)
Ex1.17+
	Replace :timer.units(x) with the new to_timeout(unit: x) for hours|minutes|seconds (This style is only applied if you're on 1.17+)


  
    
  
  Fixes


	pipes: handle pipifying when the first arg is itself a pipe: c(a |> b, d) => a |> b() |> c(d) (#214, h/t @kybishop)
	pipes: handle pipifying nested functions d(c(a |> b)) => a |> b |> c() |> d (#216, h/t @emkguts)
	with: fix a stabby with , else: (_ -> :ok) being rewritten to a case (#219, h/t @iamhassangm)


  
    
  
  1.3.3



  
    
  
  Improvements


	with do: body and variations with no arrows in the head will be rewritten to just body

	# styler:sort will sort arbitrary ast nodes within a do end block:
  Given:
  # styler:sort
  my_macro "some arg" do
    another_macro :q
    another_macro :w
    another_macro :e
    another_macro :r
    another_macro :t
    another_macro :y
  end
  We get
  # styler:sort
  my_macro "some arg" do
    another_macro :e
    another_macro :q
    another_macro :r
    another_macro :t
    another_macro :w
    another_macro :y
  end



  
    
  
  Fixes


	fix a bug in comment-movement when multiple # styler:sort directives are added to a file at the same time


  
    
  
  1.3.2



  
    
  
  Improvements


	# styler:sort can be used to sort values of key-value pairs. eg, sort the value of a single key in a map (Closes #208, h/t @ypconstante)


  
    
  
  1.3.1



  
    
  
  Improvements


	# styler:sort now works with maps and the defstruct macro


  
    
  
  Fixes


	# styler:sort no longer blows up on keyword lists :X


  
    
  
  1.3.0



  
    
  
  Improvements


# styler:sort Styler's first comment directive
Styler will now keep a user-designated list or wordlist (~w sigil) sorted as part of formatting via the use of comments. Elements of the list are sorted by their string representation.
The intention is to remove comments to humans, like # Please keep this list sorted!, in favor of comments to robots: # styler:sort. Personally speaking, Styler is much better at alphabetical-order than I ever will be.
To use the new directive, put it on the line before a list or wordlist.
This example:
# styler:sort
[:c, :a, :b]

# styler:sort
~w(a list of words)

# styler:sort
@country_codes ~w(
  en_US
  po_PO
  fr_CA
  ja_JP
)

# styler:sort
a_var =
  [
    Modules,
    In,
    A,
    List
  ]
Would yield:
# styler:sort
[:a, :b, :c]

# styler:sort
~w(a list of words)

# styler:sort
@country_codes ~w(
  en_US
  fr_CA
  ja_JP
  po_PO
)

# styler:sort
a_var =
  [
    A,
    In,
    List,
    Modules
  ]

  
    
  
  1.2.1



  
    
  
  Fixes


	|> don't pipify when the call is itself in a pipe (aka don't touch a |> b(c |> d() |>e()) |> f()) (Closes #204, h/t @paulswartz)


  
    
  
  1.2.0



  
    
  
  Improvements


	pipes: pipe-ifies when first arg to a function is a pipe. reach out if this happens in unstylish places in your code (Closes #133)
	pipes: unpiping assignments will make the assignment one-line when possible (Closes #181)
	deprecations: 1.18 deprecations	List.zip => Enum.zip
	first..last = range => first..last//_ = range




  
    
  
  Fixes


	pipes: optimizations are less likely to move comments (Closes #176)


  
    
  
  1.1.2



  
    
  
  Improvements


	Config Sorting: improve comment handling when only sorting a few nodes (Closes #187)


  
    
  
  1.1.1



  
    
  
  Improvements


	unless: rewrite unless a |> b |> c as unless !(a |> b() |> c()) rather than unless a |> b() |> c() |> Kernel.!() (h/t @gregmefford)


  
    
  
  1.1.0



  
    
  
  Improvements


The big change here is the rewrite/removal of unless due to unless "eventually" being deprecated. Thanks to @janpieper and @ypconstante for bringing this up in #190.
	unless: rewrite all unless to if (#190)
	pipes: optimize |> Stream.{each|map}(fun) |> Stream.run() to |> Enum.each(fun)


  
    
  
  Fixes


	pipes: optimizations reducing 2 pipes to 1 no longer squeeze all pipes onto one line (#180)
	if: fix infinite loop rewriting negated if with empty do body if x != y, do: (), else: :ok (#196, h/t @itamm15)


  
    
  
  1.0.0


Styler's two biggest outstanding bugs have been fixed, both related to compilation breaking during module directive organization. One was references to aliases being moved above where the aliases were declared, and the other was similarly module directives being moved after their uses in module directives.
In both cases, Styler is now smart enough to auto-apply the fixes we recommended in the old Readme.
Other than that, a slew of powerful new features have been added, the neatest one (in the author's opinion anyways) being Alias Lifting.
Thanks to everyone who reported bugs that contributed to all the fixes released in 1.0.0 as well.

  
    
  
  Improvements


Alias Lifting
Along the lines of Credo.Check.Design.AliasUsage, Styler now "lifts" deeply nested aliases (depth >= 3, ala A.B.C....) that are used more than once.
Put plainly, this code:
defmodule A do
  def lift_me() do
    A.B.C.foo()
    A.B.C.baz()
  end
end
will become
defmodule A do
  @moduledoc false
  alias A.B.C

  def lift_me do
    C.foo()
    C.baz()
  end
end
To exclude modules ending in .Foo from being lifted, add styler: [alias_lifting_exclude: [Foo]] to your .formatter.exs
Module Attribute Lifting
A long outstanding breakage of a first pass with Styler was breaking directives that relied on module attributes which Styler moved after their uses. Styler now detects these potential breakages and automatically applies our suggested fix, which is creating a variable before the module. This usually happened when folks were using a library that autogenerated their moduledocs for them.
In code, this module:
defmodule MyGreatLibrary do
  @library_options [...]
  @moduledoc make_pretty_docs(@library_options)
  use OptionsMagic, my_opts: @library_options

  ...
end
Will now be styled like so:
library_options = [...]

defmodule MyGreatLibrary do
  @moduledoc make_pretty_docs(library_options)
  use OptionsMagic, my_opts: unquote(library_options)

  @library_options library_options

  ...
end
Mix Config File Organization
Styler now organizes Mix.Config.config/2,3 stanzas according to erlang term sorting. This helps manage large configuration files, removing the "where should I put this" burden from developers AND helping find duplicated configuration stanzas.
See the moduledoc for Styler.Style.Configs for more.
Pipe Optimizations
	Enum.into(x, []) => Enum.to_list(x)
	Enum.into(x, [], mapper) => Enum.map(x, mapper)
	a |> Enum.map(m) |> Enum.join() to map_join(a, m). we already did this for join/2, but missed the case for join/1
	lhs |> Enum.reverse() |> Kernel.++(enum) => lhs |> Enum.reverse(enum)

with styles
	remove with structure with no left arrows in its head to be normal code (#174)
	with true <- x(), do: y => if x(), do: y (#173)

Everything Else
	if/unless: invert if and unless with != or !==, like we do for ! and not #132
	@derive: move @derive before defstruct|schema|embedded_schema declarations (fixes compiler warning!) #134
	strings: rewrite double-quoted strings to use ~s when there's 4+ escaped double-quotes
("\"\"\"\"" -> ~s("""")) (Credo.Check.Readability.StringSigils) #146
	Map.drop(foo, [single_key]) => Map.delete(foo, single_key) #161 (also in pipes)
	Keyword.drop(foo, [single_key]) => Keyword.delete(foo, single_key) #161 (also in pipes)


  
    
  
  Fixes


	don't blow up on def function_head_with_no_body_nor_parens (#185, h/t @ypconstante)
	fix with arrow replacement + redundant body removal creating invalid statements (#184, h/t @JesseHerrick)
	allow Kernel unary ! and not as valid pipe starts (#183, h/t @nherzing)
	fix Map.drop(x, [a | b]) registering as a chance to refactor to Map.delete

	alias: expands aliases when moving an alias after another directive that relied on it (#137)
	module directives: various fixes for unreported obscure crashes
	pipes: fix a comment-shifting scenario when unpiping
	Timex.now/1 will no longer be rewritten to DateTime.now!/1 due to Timex accepting a wider domain of "timezones" than the stdlib (#145, h/t @ivymarkwell)
	with: skip nodes which (unexpectedly) do not contain a do body (#158, h/t @DavidB59)
	then(&fun/1): fix false positives on arithmetic &1 + x / 1 (#164, h/t @aenglisc)


  
    
  
  Breaking Changes


	drop support for elixir 1.14
	ModuleDirectives: group callback attributes (before_compile after_compile after_verify) with nondirectives (previously, were grouped with use, their relative order maintained). to keep the desired behaviour, you can make new use macros that wrap these callbacks. Apologies if this makes using Styler untenable for your codebase, but it's probably not a good tool for macro-heavy libraries.
	sorting configs for the first time can change your configuration; see Mix Configs docs for more



  

    
Styler
    

[image: Hex.pm]
[image: Hexdocs.pm]
[image: Github.com]
Styler
Styler is an Elixir formatter plugin that's combination of mix format and mix credo, except instead of telling
you what's wrong, it just rewrites the code for you to fit its style rules.
You can learn more about the history, purpose and implementation of Styler from our talk: Styler: Elixir Style-Guide Enforcer @ GigCity Elixir 2023

  
    
  
  Features


Styler fixes a plethora of Elixir style and optimization issues automatically as part of mix format.
See Styler's documentation on Hex for the comprehensive list of its features.
The fastest way to see what all it can do you for you is to just try it out in your codebase... but here's a list of a few features to help you decide if you're interested in Styler:
	sorts and organizes import,alias,require and other module directives
	automatically creates aliases for repeatedly referenced modules names ("alias lifting") and makes sure aliases you've defined are being used
	keeps lists, sigils, and even arbitrary code sorted with the # styler:sort comment directive
	optimizes pipe chains for readability and performance
	rewrites deprecated Elixir standard library code, speeding adoption of new releases
	auto-fixes many credo rules, meaning you can spend less time fighting with CI

Here's another link to features Table of Contents for you

  
    
  
  Who is Styler for?


I'm just excited to be on a team that uses Styler and moves on
- Amos King

Styler was designed for a large team working in a single codebase (140+ contributors). It helps remove fiddly code review comments and linter CI slowdowns, helping our team get things done faster. Teams in similar situations might appreciate Styler.
Styler has also been extremely valuable for taming legacy Elixir codebases and general refactoring. Some of its rewrites have inspired code actions in Elixir language servers.
Conversely, Styler probably isn't a good match for:
	experimental, macro-heavy codebases
	teams that don't care about code standards


  
    
  
  Installation


Add :styler as a dependency to your project's mix.exs:
def deps do
  [
    {:styler, "~> 1.8", only: [:dev, :test], runtime: false},
  ]
end
Then add Styler as a plugin to your .formatter.exs file
[
  plugins: [Styler]
]
And that's it! Now when you run mix format you'll also get the benefits of Styler's Stylish Stylings.
Speed: Expect the first run to take some time as Styler rewrites violations of styles and bottlenecks on disk I/O. Subsequent formats won't take noticeably more time.

  
    
  
  Configuration


Styler can be configured in your .formatter.exs file
[
  plugins: [Styler],
  styler: [
    alias_lifting_exclude: [...],
    minimum_supported_elixir_version: "..."
  ]
]
	alias_lifting_exclude: a list of module names to not lift. See the Module Directive documentation for more.
	minimum_supported_elixir_version: intended for library authors; overrides the Elixir version Styler relies on with respect to some deprecation rewrites. See Deprecations documentation for more.

No Credo-Style Enable/Disable
Styler will not add configuration for ad-hoc enabling/disabling of rewrites. Sorry!
However, Smartrent has a fork of Styler named Quokka that allows for finegrained control of Styler. Maybe it's what you're looking for. If not, you can always fork it or Styler as a starting point for your own tool!
Ultimately Styler is @adobe's internal tool that we're happy to share with the world. We're delighted if you like it as is, and just as excited if it's a starting point for you to make something even better for yourself.

  
    
  
  WARNING: Styler can change the behaviour of your program!


In some cases, this can introduce bugs. It goes without saying, but look over your changes before committing to main :)
A simple example of a way Styler changes the behaviour of code is the following rewrite:
# Before: this case statement...
case foo do
  true -> :ok
  false -> :error
end

# After: ... is rewritten by Styler to be an if statement!.
if foo do
  :ok
else
  :error
end
These programs are not semantically equivalent. The former would raise if foo returned any value other than true or false, while the latter blissfully completes.
However, Styler is about style, and the if statement is (in our opinion) of much better style. If the exception behaviour was intentional on the code author's part, they should have written the program like this:
case foo do
  true -> :ok
  false -> :error
  other -> raise "expected `true` or `false`, got: #{inspect other}"
end
Also good style! But Styler assumes that most of the time people just meant the if equivalent of the code, and so makes that change. If issues like this bother you, Styler probably isn't the tool you're looking for.
Other ways Styler can change your program:
	with statement rewrites
	config file sorting
	and likely other ways. stay safe out there!


  
    
  
  Thanks & Inspiration



  
    
  
  Sourceror


Styler's first incarnation was as one-off scripts to rewrite an internal codebase to allow Credo rules to be turned on.
These rewrites were entirely powered by the terrific Sourceror library.
While Styler no longer relies on Sourceror, we're grateful for its author's help with those scripts, the inspiration
Sourceror provided in showing us what was possible, and the changes to the Elixir AST APIs that it drove.
Styler's AST-Zipper implementation in this project was forked from Sourceror. Zipper has been a crucial
part of our ability to ergonomically zip around (heh) Elixir AST.

  
    
  
  Credo


We never would've bothered trying to rewrite our codebase if we didn't have Credo rules we wanted to apply.
Credo's tests and implementations were referenced for implementing Styles that took the work the rest of the way.
Thanks to Credo & the Elixir community at large for coalescing around many of these Elixir style credos.


  

    
Style Table of Contents
    

Styler performs myriad rewrites, logically broken apart into the following groups:
	Comment Directives: leave comments to Styler instructing it to perform a task
	Control Flow Macros: styles modifying case, if, unless, cond and with statements
	Elixir Deprecations: Styles which automate the replacement or updating of code deprecated by new Elixir releases
	General Styles: general simple 1-1 rewrites that require a minimum amount of awareness of the AST
	Mix Configs: Styler applies order to chaos by organizing mix config ... stanzas
	Module Directives: Styles for alias, use, import, require, as well as alias lifting and alias application.
	Pipes styles for the famous Elixir pipe |>, including optimizations for piping standard library functions

Finally, if you're using Credo see our documentation about rules that can be disabled in Credo because Styler automatically enforces them for you, saving a modicum of CI time.


  

    
Comment Directives (# styler:sort)
    

Comment Directives are a Styler feature that let you instruct Styler to do maintain additional formatting via comments.
The plural in the name is optimistic as there's currently only one, but who knows

  
    
  
  # styler:sort


Styler can keep static values sorted for your team as part of its formatting pass. To instruct it to do so, replace any # Please keep this list sorted! notes you wrote to your teammates with # styler:sort
Sorting is done via string comparison of the code.
Styler knows how to sort the following things:
	lists of elements
	arbitrary code within do end blocks (helpful for schema-like macros)
	~w sigils elements
	keyword shapes (structs, maps, and keywords)

Since you can't have comments in arbitrary places when using Elixir's formatter,
Styler will apply those sorts when they're on the righthand side fo the following operators:
	module directives (eg @my_dir ~w(a list of things))
	assignments (eg x = ~w(a list again))
	defstruct


  
    
  
  Examples


Before
# styler:sort
[:c, :a, :b]

# styler:sort
~w(a list of words)

# styler:sort
@country_codes ~w(
  en_US
  po_PO
  fr_CA
  ja_JP
)

# styler:sort
a_var =
  [
    Modules,
    In,
    A,
    List
  ]

# styler:sort
my_macro "some arg" do
  another_macro :q
  another_macro :w
  another_macro :e
  another_macro :r
  another_macro :t
  another_macro :y
end
After
# styler:sort
[:a, :b, :c]

# styler:sort
~w(a list of words)

# styler:sort
@country_codes ~w(
  en_US
  fr_CA
  ja_JP
  po_PO
)

# styler:sort
a_var =
  [
    A,
    In,
    List,
    Modules
  ]

# styler:sort
my_macro "some arg" do
  another_macro :e
  another_macro :q
  another_macro :r
  another_macro :t
  another_macro :w
  another_macro :y
end


  

    
Control Flow Macros (<code class="inline">case</code>, <code class="inline">if</code>, <code class="inline">unless</code>, <code class="inline">cond</code>, <code class="inline">with</code>)
    

Elixir's Kernel documentation refers to these structures as "macros for control-flow".
We often refer to them as "blocks" in our changelog, which is a much worse name, to be sure.
You're likely here just to see what Styler does, in which case, please click here to skip the following manifesto on our philosophy regarding the usage of these macros.

  
    
  
  Which Control Flow Macro Should I Use?


The number of "blocks" in Elixir means there are many ways to write semantically equivalent code, often leaving developers in the dark as to which structure they should use.
We believe readability is enhanced by using the simplest api possible, whether we're talking about internal module function calls or standard-library macros.

  
    
  
  use case, if, or cond when...


We advocate for case and if as the first tools to be considered for any control flow as they are the two simplest blocks. If a branch can be expressed with an if statement, it should be. Otherwise, case is the next best choice. In situations where developers might reach for an if/elseif/else block in other languages, cond do should be used.
(cond do seems to see a paucity of use in the language, but many complex nested expressions or with statements can be improved by replacing them with a cond do).

  
    
  
  use unless when...


Never! unless is being deprecated and so should not be used.

  
    
  
  use with when...


with great power comes great responsibility
	Uncle Ben


As the most powerful of the Kernel control-flow expressions, with requires the most cognitive overhead to understand. Its power means that we can use it as a replacement for anything we might express using a case, if, or cond (especially with the liberal application of small private helper functions).
Unfortunately, this has lead to a proliferation of with in codebases where simpler expressions would have sufficed, meaning a lot of Elixir code ends up being harder for readers to understand than it needs to be.
Thus, with is the control-flow structure of last resort. We advocate that with should only be used when more basic expressions do not suffice or become overly verbose. As for verbosity, we subscribe to the Chris Keathley school of thought that judicious nesting of control flow blocks within a function isn't evil and more-often-than-not is superior to spreading implementation over many small single-use functions. We'd even go so far as to suggest that cyclomatic complexity is an inexact measure of code quality, with more than a few false negatives and many false positives.
with is a great way to unnest multiple case statements when every failure branch of those statements results in the same error. This is easily and succinctly expressed with with's else block: else (_ -> :error). As Keathley says though, Avoid Else In With Blocks. Having multiple else clauses "means that the error conditions matter. Which means that you don’t want with at all. You want case."
It's acceptable to use one-line with statements (eg with {:ok, _} <- Repo.update(changeset), do: :ok) to signify that other branches are uninteresting or unmodified by your code, but ultimately that can hide the possible returns of a function from the reader, making it more onerous to debug all possible branches of the code in their mental model of the function. In other words, ideally all function calls in a with statement head have obvious error types for the reader, leaving their omission in the code acceptable as the reader feels no need to investigate further. The example at the start of this paragraph with an Ecto.Repo call is a good example, as most developers in a codebase using Ecto are expected to be familiar with its basic API.
Using case rather than with for branches with unusual failure types can help document code as well as save the reader time in tracking down types. For example, replacing the following with a with statement that only matched against the {:ok, _} tuple would hide from readers that an atypically-shaped 3-tuple is returned when things go wrong.
case some_http_call() do
  {:ok, _response} -> :ok
  {:error, http_error, response} -> {:error, http_error, response}
end


  
    
  
  if/unless


Styler removes else: nil clauses:
if a, do: b, else: nil
# styled:
if a, do: b

  
    
  
  Negation Inversion


Styler removes negators in the head of if and unless statements by "inverting" the statement.
The following operators are considered "negators": !, not, !=, !==
Examples:
# negated `if` statement with no `else` clause are rewritten to `unless`
if not x, do: y
# Styled:
unless x, do: y

# negated `if` statements with an `else` clause have their clauses inverted and negation removed
if !x, do: y, else: z
# Styled:
if x, do: z, else: y

# negated `unless` statements are rewritten to `if`
unless x != y, do: z
# B styled:
if x == y, do: z

# `unless` with `else` is verboten; these are always rewritten to `if` statements
unless x, do: y, else: z
# styled:
if x, do: z, else: y
Because elixir relies on truthy/falsey values for its if statements, boolean casting is unnecessary and so double negation is simply removed.
if !!x, do: y
# styled:
if x, do: y

  
    
  
  case



  
    
  
  "Erlang heritage" case true/false -> if


Trivial true/false case statements are rewritten to if statements. While this results in a semantically different program, we argue that it results in a better program for maintainability. If the developer wants their case statement to raise when receiving a non-boolean value as a feature of the program, they would better serve their callers by raising something more descriptive.
In other words, Styler leaves the code with better style, trumping obscure exception design :)
# Styler will rewrite this even if the clause order is flipped,
# and if the `false` is replaced with a wildcard (`_`)
case foo do
  true -> :ok
  false -> :error
end

# styled:
if foo do
  :ok
else
  :error
end
Per the argument above, if the if statement is an incorrect rewrite for your program, we recommend this manual fix rewrite:
case foo do
  true -> :ok
  false -> :error
  other -> raise "expected `true` or `false`, got: #{inspect other}"
end

  
    
  
  cond


Styler enforces the use of true as the final clause of a cond statement when it's equivalent.
# before
cond do
  a -> b
  c -> d
  :else -> e
end
# styled
cond do
  a -> b
  c -> d
  true -> e
end

# This is left unchanged
cond do
  a -> b
  c -> d
  foo -> e
end
Styler also replaces 2-clause cond statements with if statements when possible
# Given
cond do
  a -> b
  true -> c
end
# Styled
if a do
  b
else
  c
end

# This is left unchanged, as its behaviour of raising if `foo` is falsey is assumed to be intentional
cond do
  a -> b
  foo -> c
end

  
    
  
  with


with statements are extremely expressive. Styler tries to remove any unnecessary complexity from them in the following ways.

  
    
  
  Remove Identity Else Clause


Like if statements with nil as their else clause, the identity else clause is the default for with statements and so is removed.
# Given
with :ok <- b(), :ok <- b() do
  foo()
else
  error -> error
end
# Styled:
with :ok <- b(), :ok <- b() do
  foo()
end

  
    
  
  Remove The Statement Entirely


While you might think "surely this kind of code never appears in the wild", it absolutely does. Typically it's the result of someone refactoring a pattern away and not looking at the larger picture and realizing that the with statement now serves no purpose.
Maybe someday the compiler will warn about these use cases. Until then, Styler to the rescue.
# Given:
with a <- b(),
     c <- d(),
     e <- f(),
     do: g,
     else: (_ -> h)
# Styled:
a = b()
c = d()
e = f()
g

# Given
with value <- arg do
  value
end
# Styled:
arg

  
    
  
  Replace _ <- rhs with rhs


This is another case of "less is more" for the reader.
# Given
with :ok <- x,
     _ <- y(),
     {:ok, _} <- z do
  :ok
end
# Styled:
with :ok <- x,
     y(),
     {:ok, _} <- z do
  :ok
end

  
    
  
  Replace non-branching bar <- with bar =


<- is for branching. If the lefthand side is the trivial match (a bare variable), Styler rewrites it to use the = operator instead.
# Given
with :ok <- foo(),
     bar <- baz(),
     :ok <- woo(),
     do: {:ok, bar}
# Styled
 with :ok <- foo(),
      bar = baz(),
      :ok <- woo(),
      do: {:ok, bar}

  
    
  
  Move assignments from with statement head


Just because any program could be written entirely within the head of a with statement doesn't mean it should be!
Styler moves assignments that aren't trapped between <- outside of the head. Combined with the non-pattern-matching replacement above, we get the following:
# Given
with foo <- bar,
     x = y,
     :ok <- baz,
     bop <- boop,
     :ok <- blop,
     foo <- bar,
     :success = hope_this_works! do
  :ok
end
# Styled:
foo = bar
x = y

with :ok <- baz,
     bop = boop,
     :ok <- blop do
  foo = bar
  :success = hope_this_works!
  :ok
end

  
    
  
  Remove redundant final clause


If the pattern of the final clause of the head is also the with statements do body, styler nixes the final match and makes the right hand side of the clause into the do body.
# Given
with {:ok, a} <- foo(),
     {:ok, b} <- bar(a) do
  {:ok, b}
end
# Styled:
with {:ok, a} <- foo() do
  bar(a)
end

  
    
  
  Replace with case


A with statement with a single clause in the head and an else body is really just a case statement putting on airs.
# Given:
with :ok <- foo do
  :success
else
  :fail -> :failure
  error -> error
end
# Styled:
case foo do
  :ok -> :success
  :fail -> :failure
  error -> error
end

  
    
  
  Replace with if


Given Styler rewrites trivial case to if, it shouldn't be a surprise that that same rule means that with can be rewritten to if in some cases.
# Given:
with true <- foo(), bar <- baz() do
  {:ok, bar}
else
  _ -> :error
end
# Styled:
if foo() do
  bar = baz()
  {:ok, bar}
else
  :error
end


  

    
Deprecated Elixirisms
    

Elixir's built-in formatter now does its own rewrites via the --migrate flag, but doesn't quite cover every possible automated rewrite on the hard deprecations list. Styler tries to cover the rest!
Styler will rewrite deprecations so long as their alternative is available on the given elixir version. In other words, Styler doesn't care what version of Elixir you're using when it applies the ex-1.18 rewrites - all it cares about is that the alternative is valid in your version of elixir.

  
    
  
  Configuration


While most deprecation rewrites rely on the system's Elixir version, that version can be overridden for some rewrites with the minimum_supported_elixir_version configuration. For example, to keep Styler from using rewrites that would be incompatible with Elixir 1.15:
# .formatter.exs
[
  plugins: [Styler],
  styler: [
    minimum_supported_elixir_version: "1.15.0"
  ]
]
Libraries using Styler may be running on a more modern version of Elixir while intending to support older versions. Styler can therefore break a library's minimum supported Elixir version contract when rewriting deprecated code to use more recently added standard library APIs.
For example, the to_timeout rewrite is only valid when running on Elixir 1.17 and greater. If a library supports older versions of Elixir it cannot use that function, and Styler automatically adding that function breaks them. This can be remedied by setting an earlier minimum_supported_elixir_version.
If you want to keep this configuration in sync with your project's mix.exs, consider something like the following:
# .formatter.exs
# Parse SemVer minor elixir version from project configuration
# eg `"~> 1.15"` version requirement will yield `"1.15"`
elixir_minor_version = Regex.run(~r/([\d\.]+)/, Mix.Project.config()[:elixir])

[
  plugins: [Styler],
  styler: [
    # appending `.0` to the minor version gives us a valid SemVer version string.
    minimum_supported_elixir_version: "#{elixir_minor_version}.0"
  ]
]

  
    
  
  1.20


1.20 Deprecations
No deprecation rewrites have been added to Styler for 1.20

  
    
  
  1.19


1.19 Deprecations

  
    
  
  Change Struct Updates to Map Updates


1.19 deprecates struct update syntax in favor of map update syntax.
# This
%Struct{x | y}
# Styles to this
%{x | y}
WARNING Double check your diffs to make sure your variable is pattern matching against the same struct if you want to harness 1.19's type checking features.

  
    
  
  1.18



  
    
  
  List.zip/1


# Before
List.zip(list)
# Styled
Enum.zip(list)

  
    
  
  unless


This is covered by the Elixir Formatter with the --migrate flag, but Styler brings the same transformation to codebases on earlier versions of Elixir, and insures future uses are automatically rewritten without relying on the flag.
Rewrite unless x to if !x

  
    
  
  1.17


1.17 Deprecations
	Replace :timer.units(x) with the new to_timeout(unit: x) for hours|minutes|seconds (relies on minimum_supported_elixir_version)


  
    
  
  Range Matching Without Step


# Before
first..last = range
# Styled
first..last//_ = range

# Before
def foo(x..y), do: :ok
# Styled
def foo(x..y//_), do: :ok

  
    
  
  1.16


1.16 Deprecations

  
    
  
  File.stream!/3 :line and :bytes deprecation


# Before
File.stream!(path, [encoding: :utf8, trim_bom: true], :line)
# Styled
File.stream!(path, :line, encoding: :utf8, trim_bom: true)

  
    
  
  Explicit decreasing ranges


In all these cases, the rewrite will only be applied when literals are being passed to the function. In other words, variables will not be traced back to their assignment, and so it is still possible to receive deprecation warnings on this issue.
# Before
Enum.slice(x, 1..-2)
# Styled
Enum.slice(x, 1..-2//1)

# Before
Date.range(~D[2000-01-01], ~D[1999-01-01])
# Styled
Date.range(~D[2000-01-01], ~D[1999-01-01], -1)

  
    
  
  1.15


1.15 Deprecations
	Before	After
	Logger.warn	Logger.warning
	Path.safe_relative_to/2	Path.safe_relative/2
	~R/my_regex/	~r/my_regex/
	Date.range/2 with decreasing range	Date.range/3 *
	IO.read/bin_read with :all option	replace :all with :eof



  

    
General Styles
    

(a.k.a. 1-1 or "Single Node" styles)
Function Performance & Readability Optimizations
Optimizing for either performance or readability, probably both!
These apply to the piped versions as well

  
    
  
  Strings to Sigils


Rewrites strings with 4 or more escaped quotes to string sigils with an alternative delimiter.
The delimiter will be one of " ( { | [ ' < /, chosen by which would require the fewest escapes, and otherwise preferred in the order listed.
# Before
"{\"errors\":[\"Not Authorized\"]}"
# Styled
~s({"errors":["Not Authorized"]})

  
    
  
  Large Base 10 Numbers


Style base 10 numbers with 5 or more digits to have a _ every three digits.
Formatter already does this except it doesn't rewrite "typos" like 100_000_0.
If you're concerned that this breaks your team's formatting for things like "cents" (like "$100" being written as 100_00),
consider using a library made for denoting currencies rather than raw elixir integers.
	Before	After
	10000	10_000
	1_0_0_0_0	10_000 (elixir's formatter leaves the former as-is)
	-543213	-543_213
	123456789	123_456_789
	55333.22	55_333.22
	-123456728.0001	-123_456_728.0001


  
    
  
  Enum.into -> X.new


This rewrite is applied when the collectable is a new map, keyword list, or mapset via Enum.into/2,3.
This is an improvement for the reader, who gets a more natural language expression: "make a new map from enum" vs "enumerate enum and collect its elements into a new map"
Note that all of the examples below also apply to pipes (enum |> Enum.into(...))
	Before	After
	Enum.into(enum, %{})	Map.new(enum)
	Enum.into(enum, Map.new())	Map.new(enum)
	Enum.into(enum, Keyword.new())	Keyword.new(enum)
	Enum.into(enum, MapSet.new())	MapSet.new(enum)
	Enum.into(enum, %{}, fn x -> {x, x} end)	Map.new(enum, fn x -> {x, x} end)
	Enum.into(enum, [])	Enum.to_list(enum)
	Enum.into(enum, [], mapper)	Enum.map(enum, mapper)


  
    
  
  Map/Keyword.merge w/ single key literal -> X.put


Keyword.merge and Map.merge called with a literal map or keyword argument with a single key are rewritten to the equivalent put, a cognitively simpler function.
# Before
Keyword.merge(kw, [key: :value])
# Styled
Keyword.put(kw, :key, :value)

# Before
Map.merge(map, %{key: :value})
# Styled
Map.put(map, :key, :value)

# Before
Map.merge(map, %{key => value})
# Styled
Map.put(map, key, value)

# Before
map |> Map.merge(%{key: value}) |> foo()
# Styled
map |> Map.put(:key, value) |> foo()

  
    
  
  Map/Keyword.drop w/ single key -> X.delete


In the same vein as the merge style above, [Map|Keyword].drop/2 with a single key to drop are rewritten to use delete/2
# Before
Map.drop(map, [key])
# Styled
Map.delete(map, key)

# Before
Keyword.drop(kw, [key])
# Styled
Keyword.delete(kw, key)

  
    
  
  Enum.reverse/1 and concatenation -> Enum.reverse/2


Enum.reverse/2 optimizes a two-step reverse and concatenation into a single step.
# Before
Enum.reverse(foo) ++ bar
# Styled
Enum.reverse(foo, bar)

# Before
baz |> Enum.reverse() |> Enum.concat(bop)
# Styled
Enum.reverse(baz, bop)

  
    
  
  Timex.now/0 -> DateTime.utc_now/0


Timex certainly has its uses, but knowing what stdlib date/time struct is returned by now/0 is a bit difficult!
We prefer calling the actual function rather than its rename in Timex, helping the reader by being more explicit.
This also hews to our internal styleguide's "Don't make one-line helper functions" guidance.

  
    
  
  DateModule.compare/2 -> DateModule.[before?|after?]


Again, the goal is readability and maintainability. before?/2 and after?/2 were implemented long after compare/2,
so it's not unusual that a codebase needs a lot of refactoring to be brought up to date with these new functions.
That's where Styler comes in!
The examples below use DateTime.compare/2, but the same is also done for NaiveDateTime|Time|Date.compare/2
# Before
DateTime.compare(start, end_date) == :gt
# Styled
DateTime.after?(start, end_date)

# Before
DateTime.compare(start, end_date) == :lt
# Styled
DateTime.before?(start, end_date)

  
    
  
  Implicit try


Styler will rewrite functions whose entire body is a try/do to instead use the implicit try syntax, per Credo's Credo.Check.Readability.PreferImplicitTry
# before
def foo do
  try do
    throw_ball()
  catch
    :ball -> :caught
  end
end

# Styled:
def foo do
  throw_ball()
catch
  :ball -> :caught
end

  
    
  
  Remove parenthesis from 0-arity function & macro definitions


The author of the library disagrees with this style convention :) BUT, the wonderful thing about Styler is it lets you write code how you want to, while normalizing it for reading for your entire team. The most important thing is not having to think about the style, and instead focus on what you're trying to achieve.
# Before
def foo() do
defp foo() do
defmacro foo() do
defmacrop foo() do

# Styled
def foo do
defp foo do
defmacro foo do
defmacrop foo do

  
    
  
  Variable matching on the right


# Before
case foo do
  bar = %{baz: baz? = true} -> :baz?
  opts = [[a = %{}] | _] -> a
end
# Styled:
case foo do
  %{baz: true = baz?} = bar -> :baz?
  [[%{} = a] | _] = opts -> a
end

# Before
with {:ok, result = %{}} <- foo, do: result
# Styled
with {:ok, %{} = result} <- foo, do: result

# Before
def foo(bar = %{baz: baz? = true}, opts = [[a = %{}] | _]), do: :ok
# Styled
def foo(%{baz: true = baz?} = bar, [[%{} = a] | _] = opts), do: :ok

  
    
  
  Drops superfluous = _ in pattern matching


# Before
def foo(_ = bar), do: bar
# Styled
def foo(bar), do: bar

# Before
case foo do
  _ = bar -> :ok
end
# Styled
case foo do
  bar -> :ok
end

  
    
  
  Shrink Function Definitions to One Line When Possible


# Before

def save(
       # Socket comment
       %Socket{assigns: %{user: user, live_action: :new}} = initial_socket,
       # Params comment
       params
     ),
     do: :ok

# Styled

# Socket comment
# Params comment
def save(%Socket{assigns: %{user: user, live_action: :new}} = initial_socket, params), do: :ok

  
    
  
  Assert/Refute Rewrites


Styler rewrites negated asserts, assertions that things are not nil (that's the same as just asserting the thing!) and refute with binary operators (refute's hard to reason about with anything other than == and predicates).
	before	styled
	assert !x	refute x
	assert not x	refute x
	assert !!x	assert x
	assert x != nil	assert x
	assert x == nil	no change
	assert is_nil(x)	no change
	assert !is_nil(x)	assert x
	assert x not in y	refute x in y
	refute negated	
	refute x	no change
	refute !x	assert x
	refute not x	assert x
	refute x != y	assert x == y
	refute x !== y	assert x === y
	refute x != nil	assert x == nil
	refute x not in y	assert x in y
	refute comparison	
	refute x < y	assert x >= y
	refute x <= y	assert x > y
	refute x > y	assert x <= y
	refute x >= y	assert x < y

Additionally, styler rewrites some Enum functions inside assert
	before	styled
	assert Enum.find(x, y)	assert Enum.any?(x, y)
	assert Enum.member?(y, x)	assert x in y
	assert Enum.any?(y, & &1 == x)	assert x in y
	assert Enum.any?(y, fn var -> var == x end)	assert x in y



  

    
Mix Configs
    

Mix Config files have their config stanzas sorted. Similar to the sorting of aliases, this delivers consistency to an otherwise arbitrary world, and can even help catch bugs like configuring the same key multiple times.
A file is considered a config file if
	its path matches ~r|config/.*\.exs| ~r|rel/overlays/.*\.exs|
	the file has import Config

Once a file is detected as a mix config, its config/2,3 stanzas are grouped and ordered like so:
	group config stanzas separated by assignments (x = y) together
	sort each group according to erlang term sorting
	move all existing assignments between the config stanzas to above the stanzas (without changing their ordering)


  
    
  
  THIS CAN BREAK YOUR PROGRAM


It's important to double check your configuration after running Styler on it for the first time.
First Use Advice: To limit the size of changes Styler submits to a codebase, we recommend formatting only a few (or a single) files at a time and making pull requests for each. Only commit Styler as a new formatter plugin once each of these more dangerous changes has been safely committed to the codebase.
Imagine your application configures the same value twice, once with an invalid or application breaking value, and then again with a correct value, like so:
string = "i am a string"
atom = :i_am_an_atom

config :my_app, value_must_be_an_atom: string
...
...
config :my_app, value_must_be_an_atom: atom
When styler sorts the configuration file, this dormant mistake can become a bug if the sorting changes the order such that the invalid value takes precedence (aka comes last)
string = "i am a string"
atom = :i_am_an_atom

# The value that must be an atom is now a string!
config :my_app, value_must_be_an_atom: atom
config :my_app, value_must_be_an_atom: string

  
    
  
  Examples


Sorts configs by erlang term ordering:
# Given
import Config

config :z, :x, :c
config :a, :b, :c
config :y, :x, :z
config :a, :c, :d

# Styled:
import Config

config :a, :b, :c
config :a, :c, :d

config :y, :x, :z

config :z, :x, :c
Non-config statements break the file up into chunks, where each chunk is sorted separately relative to itself.
# Given
import Config

config :z, :x, :c
config :a, :b, :c
var = "value"
config :y, :x, var
config :a, :c, var

# Styled:
import Config

config :a, :b, :c
config :z, :x, :c

var = "value"

config :a, :c, var
config :y, :x, var


  

    
Module Directives (use, alias, ...)
    


  
    
  
  Directive Organization


Modules directives are sorted into the following order:
	@shortdoc
	@moduledoc (adds @moduledoc false)
	@behaviour
	use
	import (sorted alphabetically)
	alias (sorted alphabetically)
	require (sorted alphabetically)
	everything else (order unchanged)


  
    
  
  Before


defmodule Foo do
  @behaviour Lawful
  alias A.A
  require A

  use B

  def c(x), do: y

  import C
  @behaviour Chaotic
  @doc "d doc"
  def d do
    alias X.X
    alias H.H

    alias Z.Z
    import Ecto.Query
    X.foo()
  end
  @shortdoc "it's pretty short"
  import A
  alias C.C
  alias D.D

  require C
  require B

  use A

  alias C.C
  alias A.A

  @moduledoc "README.md"
             |> File.read!()
             |> String.split("<!-- MDOC !-->")
             |> Enum.fetch!(1)
end

  
    
  
  After


defmodule Foo do
  @shortdoc "it's pretty short"
  @moduledoc "README.md"
             |> File.read!()
             |> String.split("<!-- MDOC !-->")
             |> Enum.fetch!(1)
  @behaviour Chaotic
  @behaviour Lawful

  use B
  use A.A

  import A.A
  import C

  alias A.A
  alias C.C
  alias D.D

  require A
  require B
  require C

  def c(x), do: y

  @doc "d doc"
  def d do
    import Ecto.Query

    alias H.H
    alias X.X
    alias Z.Z

    X.foo()
  end
end
If any line previously relied on an alias, the alias is fully expanded when it is moved above the alias:
# Given
alias Foo.Bar
import Bar
# Styled
import Foo.Bar

alias Foo.Bar

  
    
  
  Directive Expansion


Expands Module.{SubmoduleA, SubmoduleB} to their explicit forms for ease of searching.
# Before
import Foo.{Bar, Baz, Bop}
alias Foo.{Bar, Baz.A, Bop}

# After
import Foo.Bar
import Foo.Baz
import Foo.Bop

alias Foo.Bar
alias Foo.Baz.A
alias Foo.Bop

  
    
  
  Alias Lifting


When a module with three parts is referenced two or more times, styler creates a new alias for that module and uses it.
# Given
require A.B.C

A.B.C.foo()
A.B.C.bar()

# Styled
alias A.B.C

require C

C.foo()
C.bar()
Styler also notices when you have a module aliased and aren't employing that alias and will do the updates for you.
# Given
alias My.Apps.Widget

x = Repo.get(My.Apps.Widget, id)

# Styled
alias My.Apps.Widget

x = Repo.get(Widget, id)

  
    
  
  Collisions


Styler won't lift aliases that will collide with existing aliases, and likewise won't lift any module whose name would collide with a standard library name.
You can specify additional modules to exclude from lifting via the :alias_lifting_exclude configuration option. For the example above, the following configuration would keep Styler from creating the alias A.B.C node:
# .formatter.exs
[
  plugins: [Styler],
  styler: [alias_lifting_exclude: [:C]],
]

  
    
  
  Alias Application


Styler applies aliases in those cases where a developer wrote out a full module name without realizing that the module is already aliased.
# Given
alias A.B
alias A.B.C
alias A.B.C.D, as: X

A.B.foo()
A.B.C.foo()
A.B.C.D.woo()
C.D.woo()

# Styled
alias A.B
alias A.B.C
alias A.B.C.D, as: X

B.foo()
C.foo()
X.woo()
X.woo()

  
    
  
  Adds @moduledoc false


Adds @moduledoc false to modules without a moduledoc.
Styler does this for two reasons:
	Its appearance during code review is a reminder to the author and reviewer that they may want to document the module. This can otherwise be easily forgotten.

	To normalize the use of @moduledoc false, because it's preferable to docstrings which convey no actual information.
 For example: @moduledoc "The module for functions for interacting with Widgets" in the Widget module is crueler code to have written than just @moduledoc false, because including a string gave the reader hope that the author was going to help them comprehend the module. That false hope causes more harm than just saying "Sorry, you're on your own." (aka @moduledoc false)


In conformance with the precedent set by Credo, this @moduledoc is not added if the module's name ends with any of the following:
	Test
	Mixfile
	MixProject
	Controller
	Endpoint
	Repo
	Router
	Socket
	View
	HTML
	JSON



  

    
Pipe Chains
    


  
    
  
  Pipe Start


Styler will ensure that the start of a pipechain is a 0-arity function, a raw value, or a variable.
Enum.at(enum, 5)
|> IO.inspect()

# Styled:
enum
|> Enum.at(5)
|> IO.inspect()
If the start of a pipe is a block expression, styler will create a new variable to store the result of that expression and make that variable the start of the pipe.
if a do
  b
else
  c
end
|> Enum.at(4)
|> IO.inspect()

# Styled:
if_result =
  if a do
    b
  else
    c
  end

if_result
|> Enum.at(4)
|> IO.inspect()

  
    
  
  Add parenthesis to function calls in pipes


a |> b |> c |> d
# Styled:
a |> b() |> c() |> d()

  
    
  
  then/2 improvements



  
    
  
  Add then/2 when defining and calling anonymous functions in pipes


a |> (fn x -> x end).() |> c()
# Styled:
a |> then(fn x -> x end) |> c()

  
    
  
  Remove Redundant then/2


When the piped argument is being passed as the first argument to the inner function, there's no need for then/2.
a |> then(&f(&1, ...)) |> b()
# Styled:
a |> f(...) |> b()

  
    
  
  Piped function optimizations


Two function calls into one! Fewer steps is always nice.
# reverse |> concat => reverse/2
a |> Enum.reverse() |> Enum.concat(enum) |> ...
# Styled:
a |> Enum.reverse(enum) |> ...

# filter |> count => count(filter)
a |> Enum.filter(filterer) |> Enum.count() |> ...
# Styled:
a |> Enum.count(filterer) |> ...

# map |> join => map_join
a |> Enum.map(mapper) |> Enum.join(joiner) |> ...
# Styled:
a |> Enum.map_join(joiner, mapper) |> ...

# Enum.map |> X.new() => X.new(mapper)
# where X is one of: Map, MapSet, Keyword
a |> Enum.map(mapper) |> Map.new() |> ...
# Styled:
a |> Map.new(mapper) |> ...

# Enum.map |> Enum.into(empty_collectable) => X.new(mapper)
# Where empty_collectable is one of `%{}`, `Map.new()`, `Keyword.new()`, `MapSet.new()`
# Given:
a |> Enum.map(mapper) |> Enum.into(%{}) |> ...
# Styled:
a |> Map.new(mapper) |> ...

# Given:
a |> b() |> Stream.each(fun) |> Stream.run()
a |> b() |> Stream.map(fun) |> Stream.run()
# Styled:
a |> b() |> Enum.each(fun)
a |> b() |> Enum.each(fun)

# Given:
a |> Enum.filter(fun) |> List.first() |> ...
a |> Enum.filter(fun) |> List.first(default) |> ...
# Styled:
a |> Enum.find(fun) |> ...
a |> Enum.find(default, fun) |> ...

  
    
  
  Unpiping Single Pipes


Styler rewrites pipechains with a single pipe to be function calls. Notably, this rule combined with the optimizations rewrites above means some chains with more than one pipe will also become function calls.
foo = bar |> baz()
# Styled:
foo = baz(bar)

map = a |> Enum.map(mapper) |> Map.new()
# Styled:
map = Map.new(a, mapper)

  
    
  
  Pipe-ify


If the first argument to a function call is a pipe, Styler makes the function call the final pipe of the chain.
d(a |> b |> c)
# Styled
a |> b() |> c() |> d()
Styler does not pipe-ify nested function calls if there are no pipes:
# Styler does not change this
d(c(b(a))
If you want Styler to do the work of transforming nested function calls into a pipe for you, change the innermost function call to a pipe, then format. Styler will take care of the rest.
# To have Styler change this to pipes
d(c(b(a))
# You will have to add a single pipe to the innermost function call, like so:
d(c(a |> b))
# At which point Styler will pipe-ify the entire chain
a |> b() |> c() |> d()

